CUMULATIVE PRODUCT

PURPOSE

Compute the cumulative (or partial) product for a variable. That is, for element I the cumulative product is the product of elements 1 through I.

SYNTAX

LET <y2> = CUMULATIVE PRODUCT <y1> <SUBSET/EXCEPT/FOR qualification>
where <y1> is a response variable;
<y2> is a variable where the computed cumulative products are stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

LET Y2 = CUMULATIVE PRODUCT Y1
LET Y2 = CUMULATIVE PRODUCT Y1 FOR I = 1 1 30

NOTE

Although this is an infrequently used command, it can be useful for the following type problem. Suppose you have an input variable to be used in further calculations. However, you only want to use those values up to the first occurrence of a value greater than some cut-off value. Do the following:

```
LET CUTOFF = <supply value>
LET N = SIZE Y1
LET TAG = 1 FOR I = 1 1 N
LET TAG = 0 SUBSET Y1 > CUTOFF
LET TAG = CUMULATIVE PRODUCT TAG
```

Now TAG contains 1's up to the first value greater than the cutoff and all 0's after that. This TAG variable can be used as a SUBSET variable in subsequent commands.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

SUM	=	Compute the sum of a variable.
CUMULATIVE SUM	=	Compute the cumulative sum of a variable.
CUMULATIVE INTEGRAL	=	Compute the cumulative integral of a variable.

APPLICATIONS

Mathematics

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```
. Find the first random number that is greater than 2.5
LET Y1 = NORMAL RANDOM NUMBERS FOR I = 1 1 100
LET CUTOFF = 2.5
LET N = SIZE Y1
LET TAG = 1 FOR I = 1 1 N
LET TAG = 0 SUBSET Y1 > CUTOFF
LET TAG = CUMULATIVE PRODUCT TAG
LET ID = SUM TAG
LET ID = ID + 1
```